In[1]:=
```
SetDirectory["~/KappaLib/"];
<< kappaLib-1.2.m
```

```
Loading KappaLib v1.2
```

- **Verify that the Fresnel polynomial for the SDCM medium class in**

  Lindell, Bergamin, Favaro : Decomposable medium conditions in four - dimensional representation

  factorises into the product of two quadratic forms.

- **Define sigma as the most general (2,2)-tensor with only a skewon part**

In[3]:=
```
(*
   Any kappa with only a skewon part can be written as

      S /\ Id + Id /\ S

   for a trace-free (1,1)-tensor S and the identity (1,1)-tensor Id.

   See PDCM notebook, and for the emPQToKappa routine, see the KappaLib source code
*)
S = emMatrix["sh", 4, Structure → "General"];
S = S - 1 / 4 Tr[S] IdentityMatrix[4];
id = IdentityMatrix[4];
sigma = Simplify[emPQToKappa[id, S] + emPQToKappa[S, id]];

(* Verify: medium has only a skewon part *)
Union[Flatten[Simplify[sigma + emPoincare[sigma]]]]
```

Out[7]=
```
{0}
```

- **Other parameters in SDCM medium**

In[8]:=
```
AA = emMatrix["a", 4, Structure → "AntiSymmetric"];
BB = emMatrix["b", 4, Structure → "AntiSymmetric"];
(* rho = scalar density of weight 1 *)
(* C1 = constant *)
```

- **Define kappa as most general SDCM medium**

In[10]:=
```
kappa =
  C1 emIdentityKappa[] + emBiProduct[rho, AA, BB] + emBiProduct[rho, BB, AA] + sigma;
```

- **Check that Fresnel polynomial factorises into product of two quadratic forms**

In[11]:=
```
vars = {x0, x1, x2, x3};
fresnel = emKappaToFresnel[kappa, vars];
fresnelExp = 4 rho (vars.S.AA.vars) (vars.S.BB.vars);
Simplify[fresnel - fresnelExp]
```

Out[14]=
```
0
```